Étude d'une courbe du second degré

Nous allons effectuer la *réduction* de la courbe d'équation $2x^2+3xy+y^2-4x-5y=0$ en utilisant ***Mathematica*** en appui dans les calculs.

```
equ1 = 2x^2+3x y +y^2-4x-5y
```

$2x^2+3xy-4x+y^2-5y$

Pour avoir une idée de la courbe à détecter, nous pouvons la représenter avec `ContourPlot`.

```
ContourPlot[equ1 == 0,{x,-30,100},{y,-100,30}]
```



Identifions les coefficients α, β et γ.

```
{alpha,beta,gamma} = Coefficient[equ1,{x^2,x y, y^2}]
```

$\{2,3,1\}$

Calculons le discriminant de la courbe.

```
Delta = beta^2-4alpha gamma
```

1

Elle est du genre hyperbole, cela se confirme ! Déterminons l'angle de la rotation à effectuer pour éliminer le terme *rectangle*.

```
theta = 1/2 ArcTan[beta/(alpha-gamma)]
```

$\frac{1}{2}\arctan(3)$

Calculons $\cos\theta$ et $\sin\theta$. On notera que sans incitation particulière, ***Mathematica***, dans ce cas, ne fais aucune simplication...

Nous lui demandons donc de passer aux exponentielles et de simplifier au maximum le résultat obtenu !

```
cos = Cos[theta] // TrigToExp // FullSimplify
```

$\frac{1}{2}\sqrt{2+\sqrt{\frac{2}{5}}}$

```
sin = Sin[theta] // TrigToExp // FullSimplify
```

$\frac{1}{2}\sqrt{2-\sqrt{\frac{2}{5}}}$

Nous sommes maintenant en mesure d'écrire les règles de substitution correspondantes à notre changement de repère par rotation.

```
rotation = {x->cos X - sin Y,y->sin X + cos Y}
```

$$\left\{x\to\frac{1}{2}\sqrt{2+\sqrt{\frac{2}{5}}}X-\frac{1}{2}\sqrt{2-\sqrt{\frac{2}{5}}}Y, y\to\frac{1}{2}\sqrt{2-\sqrt{\frac{2}{5}}}X+\frac{1}{2}\sqrt{2+\sqrt{\frac{2}{5}}}Y\right\}$$

Voici la nouvelle équation.

```
equ2 = equ1 /.rotation // FullSimplify
```

$$\frac{1}{10}\left(5\left(3+\sqrt{10}\right)X^2-\sqrt{2050+555\sqrt{10}}X-Y\left(5\left(\sqrt{10}-3\right)Y+\sqrt{2050-555\sqrt{10}}\right)\right)$$

On vérifie bien la disparition du terme rectangle. Nous allons maintenant préparer le changement de repère par translation.

L'équation précédente est de la forme $\lambda X^2+\mu Y^2+AX+BY=0$.

```
{lambda,mu,A,B} = Coefficient[equ2,{X^2,Y^2,X,Y}]
```

$$\left\{\frac{1}{2}\left(3+\sqrt{10}\right),\frac{1}{10}\left(15-5\sqrt{10}\right),-\frac{1}{10}\sqrt{2050+555\sqrt{10}},-\frac{1}{10}\sqrt{2050-555\sqrt{10}}\right\}$$

Je vous laise retrouver les formules de changement de repère qui vont *digérer* les termes en X et Y.

```
translation = {X->x-A/(2lambda),Y->y-B/(2mu)}
```

$$\left\{X\to x+\frac{\sqrt{2050+555\sqrt{10}}}{10\left(3+\sqrt{10}\right)}, Y\to y+\frac{\sqrt{2050-555\sqrt{10}}}{2\left(15-5\sqrt{10}\right)}\right\}$$

Effectuons la substitution.

```
equ3 = equ2 /. translation // FullSimplify
```

$$\frac{1}{2}\left(\left(3+\sqrt{10}\right)x^2-\left(\sqrt{10}-3\right)y^2+12\right)$$

L'utilisation de `Collect` ci-dessous permet de disposer d'une écriture compacte des coefficients des monômes.

```
equ3 = Collect[equ3/6,{x,y}]
```

$$\frac{1}{12}\left(3+\sqrt{10}\right)x^2+\frac{1}{12}\left(3-\sqrt{10}\right)y^2+1$$

Un rapide coup d'œil, nous permet de constater que l'équation précédente peut se mettre sous la forme :

$$\frac{y^2}{b^2}-\frac{x^2}{a^2}=1.$$

Avec a et b qui sont :

```
a = Sqrt[1/Coefficient[equ3,x^2]] // Simplify
```

$2\sqrt{3\left(\sqrt{10}-3\right)}$

☐ `b = Sqrt[-1/Coefficient[equ3,y^2]] // Simplify`

$2\sqrt{3\left(3+\sqrt{10}\right)}$

Calculons le coefficient c de l'hyperbole ainsi que l'excentricité.

☐ `c = Sqrt[a^2+b^2] // Simplify`

$2 2^{3/4}\sqrt{3}\sqrt[4]{5}$

☐ `e = c / b // Simplify`

$\frac{2^{3/4}\sqrt[4]{5}}{\sqrt{3+\sqrt{10}}}$

Numériquement :

☐ `{a,b,c,e} // N`

$\{1.39547, 8.59926, 8.71175, 1.01308\}$

L'excentricité est proche de 1, ce qui explique le caractère applati des branches de l'hyperbole.